

Decomposing Montagovian Dynamics

Greg Kobele

May 5, 2016

1 Introduction

de Groot [2006] demonstrates how to give a dynamic interpretation to formulae in higher order logic, not by, as is done in most works on dynamic semantics, giving a non-standard model-theoretic interpretation of such formulae, but rather by giving a systematic syntactic translation of these formulae into other formulae which, when interpreted in the standard way, are equivalent to the dynamic interpretation of the original formulae. This is achieved by lifting the types of expressions to be functions from left contexts (representing information about already established discourse referents) to functions from right contexts (representing discourse continuations) to truth values.

The goal of this short note is to show that the dynamic translation of de Groot [2006] can be factored into the composition of two simpler parts; a left-context translation and a right-context translation, or, more precisely, a semantics dealing with dynamism without discourse referents, and a semantics dealing with discourse referents without dynamism.

This is a literate Coq document, with source code available on the web at <https://github.com/gkobele/decomposing-montague>.

2 The formal language

I begin by defining the basic structures of the language. This will serve both as a metalanguage (the target language of the translations) and also as an object language (the source languages of the translations).

The basic idea of a translation procedure is that an uninterpreted object language is translated into a language that is already understood, the meta-language. The same syntactic forms will be used (a simple fragment of the lambda calculus) for both. This offers a sort of *modularity* of semantic description. From the semanticist's perspective, this means that certain aspects of meaning can be systematically ignored, knowing that the semantic terms actually written down are (via the translation process) notational shorthands for richer meaning representations. More speculatively, one might conjecture that each successive translation might have some sort of deeper reality, perhaps recapitulating the stages of a

\top	<i>true</i>	<i>farmer</i>	<i>p0</i>
\perp	<i>false</i>	<i>donkey</i>	<i>p1</i>
\neg	<i>not</i>	<i>laugh</i>	<i>p2</i>
\wedge	<i>and</i>	<i>bray</i>	<i>p3</i>
\vee	<i>or</i>	<i>own</i>	<i>r0</i>
\rightsquigarrow	<i>imply</i>	<i>beat</i>	<i>r1</i>

Figure 1: Abbreviations for constants

learner’s semantic language acquisition, or perhaps even as reflective of different levels of deep semantic analysis of a sentence made as needed during language processing.

The language is typed, with two basic types; e and t . Nothing is assumed about expressions of type e ; in particular, no expression of the language will have a type ending in e . In contrast, the type t , is viewed as the type of complete propositions, is the locus of recursion in this language. Thus, *and* is a binary term forming operator. There are infinitely many inhabitants of type t , many of which are intuitively semantically equivalent. There are constants *true* and *false*, boolean connectives *and*, *or* and *not*, term forming operators $p\ n$ and $r\ n$ (the n th unary and binary relations respectively), and generalized quantifiers *some*, *all* and *pro* (representing a pronoun). These provide the basic alphabet for building semantic terms (of type t). With the exception of *pro*, these can be interpreted in the usual way in higher order models. *pro* is not (intended to be understood as) a term in the meta-language, and will only be given a meaning during the translation process. [de Groote \[2006\]](#) also has a constant *who*, which is however definable as usual in terms of property conjunction: $\mathbf{who}\ P\ Q = \lambda x.Px \wedge Qx$.

```

Inductive t : Set :=
| true : t
| false : t
| p : nat → e → t
| r : nat → e → e → t
| and : t → t → t
| or : t → t → t
| imply : t → t → t
| not : t → t
| some : (e → t) → t
| all : (e → t) → t
| pro : (e → t) → t.

```

To make terms more readable, I introduce familiar notation, as shown in figure 1.

Example *Every_donkey_brayed* : $\mathbf{all}\ (\mathbf{fun}\ x \Rightarrow \mathbf{donkey}\ x \rightsquigarrow \mathbf{bray}\ x) = \mathbf{all}\ (\mathbf{fun}\ x \Rightarrow (p\ 1\ x) \rightsquigarrow (p\ 3\ x))$.

The intended interpretation of terms of type t is of course in an algebra (boolean or heyting) where \top is understood as truth (the top element of the algebra), and \perp as falsity (the

bottom element of the algebra). The connective symbols are to be understood as appropriate operations in the algebra (meet, join, and complement if in a boolean algebra, and meet, join and relative complement in a heyting algebra). Although this setup is independent of one's particular choice of logic (classical or intuitionistic), classical logic is more familiar in linguistics, and generates a coarser equivalence relation over terms. I will write $\phi \equiv \psi$ to indicate that ϕ is classically equivalent to ψ . The notation $i \hat{\wedge} n o$ abbreviates the type $\underbrace{i \rightarrow \dots \rightarrow i}_{n \text{ times}} \rightarrow o$.

3 Dynamism

In this section, dynamism is introduced to the system above. Although the dynamic properties of connectives are typically argued for based on the behaviour of pronouns, the statement of the dynamic properties of connectives needn't make reference to pronouns, which is exploited here.

The intuitive understanding of the dynamic meaning of a sentence is as explicating its contribution to the discourse; it specifies how the meaning of the upcoming discourse is influenced by its own meaning. Accordingly the dynamic type Ω of a sentence is $t \rightarrow t$; a function which modifies the meaning of an upcoming discourse. Dynamic translations of terms of type t into terms of type Ω are now defined. These terms of type Ω are intended to capture in an intuitive way the *dynamic* meaning of the original terms of type t .

Atomic sentences, qua n-ary predicates, are simply conjoined with the discourse continuation. That is, the meaning of a discourse consisting of only atomic sentences is hereby stipulated to be their conjunction.

```
Fixpoint atom_d_tr (n : nat) : (e ^ n) t → (e ^ n) Ω :=
  match n with
  | 0 ⇒ fun atom φ ⇒ atom ∧ φ
  | S m ⇒ fun atom x ⇒ atom_d_tr m (atom x)
  end.
```

Example *dynamic_beats* : $atom_d_tr\ 2\ beat = fun\ x\ y\ \phi \Rightarrow beat\ x\ y \wedge \phi$.

The atomic sentences \top and \perp , representing the always true and always false sentences, obtain from the general treatment of atomic sentences above the meanings $fun\ \phi \Rightarrow \top \wedge \phi$ and $fun\ \phi \Rightarrow \perp \wedge \phi$ respectively, which are semantically equivalent to the identity function and the constant false function respectively. To simplify the statement of the main result, the translations of \top and \perp are simply taken to be these simpler functions.

Definition *true_d_tr* : $\Omega := fun\ \phi \Rightarrow \phi$.

Example *true_id* : $\forall\ \phi, atom_d_tr\ 0\ \top\ \phi \equiv true_d_tr\ \phi$.

Definition *false_d_tr* : $\Omega := fun\ _ \Rightarrow \perp$.

Example *false_false* : $\forall\ \phi, atom_d_tr\ 0\ \perp\ \phi \equiv false_d_tr\ \phi$.

Connectives are classified as externally dynamic just in case a discourse referent introduced internally to them is accessible externally. They are internally dynamic just in case a discourse referent introduced inside of one conjunct is accessible in the other.

As conjunction is externally dynamic, the discourse continuation of a conjunction should be in the scope of both conjuncts. As it is also internally dynamic, the second conjunct should be in the scope of the first. The discourse continuation of the first conjunct P is thus the result of continuing the second conjunct Q with the discourse continuation of the entire coordination, ϕ . In other words, a discourse $P \wedge Q$. D is interpreted as P . Q . D .

Definition *and_d_tr* : $\Omega \rightarrow \Omega \rightarrow \Omega :=$
 $\text{fun } (P \ Q : \Omega) (\phi : t) \Rightarrow P \ (Q \ \phi)$.

Negation is externally static (i.e. not externally dynamic); so the discourse continuation of a negated expression P must not be in its scope. Instead, the negated expression P is given a trivial discourse continuation, \top , and its negation is treated as an atomic proposition; i.e. it is conjoined with the remainder of the discourse.

Definition *not_d_tr* : $\Omega \rightarrow \Omega :=$
 $\text{fun } (P : \Omega) \Rightarrow \text{atom_d_tr } 0 \ (\neg (P \ \top))$.

Example *not_d_tr_externally_static* :
 $\forall (P : \Omega) (\phi : t), \text{not_d_tr } P \ \phi = \neg (P \ \top) \wedge \phi$.

The other dynamic connectives will be defined in terms of dynamic conjunction and dynamic negation. Dynamic behaviour, whether internal or external, will implicate a dynamic conjunction, and static behaviour a dynamic negation.

Implication is externally static, but internally dynamic from antecedent P to consequent Q . This is captured by its classically valid reformulation in terms of \wedge and \neg : $P \rightarrow Q \equiv \neg(P \wedge \neg Q)$

Definition *imp_d_tr* : $\Omega \rightarrow \Omega \rightarrow \Omega :=$
 $\text{fun } (P \ Q : \Omega) \Rightarrow \text{not_d_tr } (\text{and_d_tr } P \ (\text{not_d_tr } Q))$.

The dynamic behaviour of disjunction is less clear [Groenendijk and Stokhof, 1991]. Sentences like: “Either this house doesn’t have a bathroom, or it is hidden” suggest that disjunction should be (at least) internally dynamic, however it is common to assume that disjunction is in fact completely static. This is in fact predicted by its classical reformulation in terms of negation and conjunction: $P \vee Q \equiv \neg(\neg P \wedge \neg Q)$

Definition *or_d_tr* : $\Omega \rightarrow \Omega \rightarrow \Omega :=$
 $\text{fun } (P \ Q : \Omega) \Rightarrow \text{not_d_tr } (\text{and_d_tr } (\text{not_d_tr } P) \ (\text{not_d_tr } Q))$.

The generic treatment of dynamic GQs has it that the discourse continuation of the entire sentence is moved into the scope of the GQ.

Definition *quant_d_tr* ($q : (e \rightarrow t) \rightarrow t$) : $(e \rightarrow \Omega) \rightarrow \Omega :=$
 $\text{fun } (P : e \rightarrow \Omega) (\phi : t) \Rightarrow q \ (\text{fun } x \Rightarrow P \ x \ \phi)$.

Exceptionally, from this perspective, universal quantification is externally static, which is captured by its de Morgan reformulation in terms of existential quantification: $\forall x.Px \equiv \neg \exists x.\neg Px$

Definition $all_d_tr : (e \rightarrow \Omega) \rightarrow \Omega :=$

$\text{fun } (P : e \rightarrow \Omega) \Rightarrow \text{not_d_tr } (\text{quant_d_tr } \mathbf{some} \ (\text{fun } x \Rightarrow \text{not_d_tr } (P \ x))).$

These definitions allow for a concise statement of the translation d_tr (mnemonic for *dynamic* translation) from terms of type t , with their dynamic potential *implicit* into terms of type Ω with their dynamic meanings made *explicit*. We see that the translation d_tr is just a homomorphism.

Fixpoint $d_tr (\phi : t) : \Omega :=$

```

match  $\phi$  with
|  $\top \Rightarrow \text{true\_d\_tr}$ 
|  $\perp \Rightarrow \text{false\_d\_tr}$ 
|  $p \ n \ x \Rightarrow \text{atom\_d\_tr } 1 \ (p \ n) \ x$ 
|  $r \ n \ x \ y \Rightarrow \text{atom\_d\_tr } 2 \ (r \ n) \ x \ y$ 
|  $\neg \psi \Rightarrow \text{not\_d\_tr } (d\_tr \ \psi)$ 
|  $\psi \wedge \chi \Rightarrow \text{and\_d\_tr } (d\_tr \ \psi) \ (d\_tr \ \chi)$ 
|  $\psi \vee \chi \Rightarrow \text{or\_d\_tr } (d\_tr \ \psi) \ (d\_tr \ \chi)$ 
|  $\psi \rightsquigarrow \chi \Rightarrow \text{imp\_d\_tr } (d\_tr \ \psi) \ (d\_tr \ \chi)$ 
|  $\text{pro } P \Rightarrow \text{quant\_d\_tr } \text{pro} \ (\text{fun } x \Rightarrow d\_tr \ (P \ x))$ 
|  $\mathbf{some} \ P \Rightarrow \text{quant\_d\_tr } \mathbf{some} \ (\text{fun } x \Rightarrow d\_tr \ (P \ x))$ 
|  $\mathbf{all} \ P \Rightarrow \text{all\_d\_tr } (\text{fun } x \Rightarrow d\_tr \ (P \ x))$ 

```

end.

This translation, as can be seen by the examples to follow, allows existentials to dynamically extend their scope over upcoming sentences, while disallowing universals from doing the same.

Example $Some_farmer_laughed :$

```

 $\forall \phi : t,$ 
 $d\_tr \ (\mathbf{some} \ (\text{fun } x \Rightarrow \text{farmer } x \wedge \text{laugh } x)) \ \phi$ 
 $\equiv \mathbf{some} \ (\text{fun } x \Rightarrow \text{farmer } x \wedge \text{laugh } x \wedge \phi).$ 

```

Example $Every_farmer_laughed :$

```

 $\forall \phi : t,$ 
 $d\_tr \ (\mathbf{all} \ (\text{fun } x \Rightarrow \text{farmer } x \rightsquigarrow \text{laugh } x)) \ \phi$ 
 $\equiv (\mathbf{all} \ (\text{fun } x : e \Rightarrow \text{farmer } x \rightsquigarrow \text{laugh } x)) \wedge \phi.$ 

```

4 Contexts

Here we present an interpretation of context-independent sentences into context *dependent* ones. This does not presuppose (or introduce) dynamism.

Following [de Groote \[2006\]](#), we introduce a new type γ for contexts. This type is *abstract*, in the sense that its properties will be given axiomatically; *anything* that satisfies these properties may be used as a context. We require only that we can interact with objects

of type γ in two ways. First, that a context may be updated with an individual to obtain a new context.

Variable *update* : $e \rightarrow \gamma \rightarrow \gamma$.

The notation $x :: c$ will abbreviate the more cumbersome *update* x c .

And second, that salient individuals may be retrieved from a context.

Variable *sel* : $\gamma \rightarrow e$.

In particular, *sel* should be thought of as a placeholder for one's favourite pronoun resolution algorithm; in this view, the job of semantics is to give the appropriate input to the pronoun resolution algorithm, but the inner workings of this algorithm are semantically opaque. [de Groot](#) [2006] takes for concreteness γ to be the type of lists of individuals, updating to be achieved by adding an individual to a list, and selection to be retrieving some element of a list.

Terms of type t will be viewed as having implicit contexts. We will define a translation from terms of type t which makes their contexts, context updating, and context passing explicit.

Contexts are incorporated by adding a context parameter to every atomic type.

Definition $G := \gamma \rightarrow t$.

Definition $E := \gamma \rightarrow e$.

Thus a term of type t will be translated into one of type G , and one of type e will be translated into one of type E .

Inherently context insensitve n -ary connectives, functions, and predicates can be lifted to context sensitive ones by simply distributing the context to their context-sensitive arguments.

Fixpoint $distr_g_tr \{A B : \text{Type}\} (n : \text{nat}) : (\gamma \rightarrow (A \wedge\wedge n) B) \rightarrow ((\gamma \rightarrow A) \wedge\wedge n) (\gamma \rightarrow B) :=$
 $\text{match } n \text{ as } n0 \text{ return } (\gamma \rightarrow (A \wedge\wedge n0) B) \rightarrow ((\gamma \rightarrow A) \wedge\wedge n0) (\gamma \rightarrow B) \text{ with}$
 $| O \Rightarrow \text{fun } xx \Rightarrow xx$
 $| S m \Rightarrow \text{fun } xx \phi \Rightarrow distr_g_tr m (\text{fun } c : \gamma \Rightarrow xx c (\phi c))$
 end.

Definition $lift_g_tr \{A B : \text{Type}\} (n : \text{nat}) (m : (A \wedge\wedge n) B) : ((\gamma \rightarrow A) \wedge\wedge n) (\gamma \rightarrow B) :=$
 $@distr_g_tr A B n (\text{fun } _ \Rightarrow m).$

Example *contextual_individual* $(j : e) : lift_g_tr (A := e) 0 j = (\text{fun } _ \Rightarrow j).$

Example *contextual_beats* : $lift_g_tr 2 \text{beat} = \text{fun } x y c \Rightarrow \text{beat } (x c) (y c).$

A pronoun *pro* is translated as a generalized quantifier generated from the (context-sensitive) individual *sel*.

Definition $pro_g_tr : (E \rightarrow G) \rightarrow G := \text{fun } P \Rightarrow P \text{sel}.$

The individual argument to the property of a GQ is a lifted context-insensitive variable. The context passed to this property is enriched with the unlifted variable.

Definition $quant_g_tr$ ($gq : (e \rightarrow t) \rightarrow t$) : $(E \rightarrow G) \rightarrow G :=$
 $\text{fun } P (c : \gamma) \Rightarrow gq (\text{fun } x \Rightarrow P (\text{lift_g_tr } (A := e) 0 x) (x :: c)).$

The function g_tr (mnemonic for *gamma* translation) makes explicit the implicit context manipulation in terms of type t .

Fixpoint g_tr ($\phi : t$) : $G :=$

match ϕ **with**

| $\top \Rightarrow \text{lift_g_tr } (A := t) 0 \top$

| $\perp \Rightarrow \text{lift_g_tr } (A := t) 0 \perp$

| $\neg \psi \Rightarrow \text{lift_g_tr } 1 \neg (g_tr \psi)$

| $\psi \wedge \chi \Rightarrow \text{lift_g_tr } 2 \wedge (g_tr \psi) (g_tr \chi)$

| $\psi \rightsquigarrow \chi \Rightarrow \text{lift_g_tr } 2 \rightsquigarrow (g_tr \psi) (g_tr \chi)$

| $\psi \vee \chi \Rightarrow \text{lift_g_tr } 2 \vee (g_tr \psi) (g_tr \chi)$

| $p n a \Rightarrow \text{lift_g_tr } 1 (p n) (\text{lift_g_tr } (A := e) 0 a)$

| $r n a b \Rightarrow \text{lift_g_tr } 2 (r n) (\text{lift_g_tr } (A := e) 0 a) (\text{lift_g_tr } (A := e) 0 b)$

| **some** $P \Rightarrow \text{quant_g_tr some } (\text{fun } (x : E) (c : \gamma) \Rightarrow g_tr (P (x c)) c)$

| **all** $P \Rightarrow \text{quant_g_tr all } (\text{fun } (x : E) (c : \gamma) \Rightarrow g_tr (P (x c)) c)$

| **pro** $P \Rightarrow \text{pro_g_tr } (\text{fun } (x : E) (c : \gamma) \Rightarrow g_tr (P (x c)) c)$

end.

In the example below, which is the translation of the sentence *he laughed* into context-sensitive terms, note that the referent of the pronoun *he* must be found in the context of utterance c .

Example $he_laughed$:

$\forall (c : \gamma), g_tr (\text{pro laugh}) c = \text{laugh } (sel c).$

In the following example, the translation of the sentence *some farmer owned it*, note that the referent of *it* is to be found in the context of utterance c which has been updated with a discourse referent x which is a farmer. Of course, for multiple reasons (the pronoun is incompatible with the animacy of the farmer, and the pronoun is syntactically too local) the pronoun should not in this case be resolved to the discourse referent x .

Example $some_farmer_owned_it$:

$\forall (c : \gamma), g_tr (\text{some } (\text{fun } x \Rightarrow \text{farmer } x \wedge (\text{pro } (\text{fun } y \Rightarrow \text{own } y x)))) c$
 $= \text{some } (\text{fun } x : e \Rightarrow \text{farmer } x \wedge \text{own } (sel (x :: c)) x).$

5 De Groote's Montagovian Dynamics

The type of a sentence ω is defined to be $\gamma \rightarrow (\gamma \rightarrow t) \rightarrow t$; a function from a context γ to the type $(\gamma \rightarrow t) \rightarrow t$ of the *continuation* of a context.

Atomic predicates do not modify their context; they simply conjoin their static meaning with the meaning of the discourse continuation in the context.

Definition *pred_dg_tr* ($n : nat$) : $e \rightarrow \omega :=$
 $\text{fun } (x : e) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow (p \ n \ x) \wedge (\phi \ c).$

Definition *rel_dg_tr* ($n : nat$) : $e \rightarrow e \rightarrow \omega :=$
 $\text{fun } (x \ y : e) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow (r \ n \ x \ y) \wedge (\phi \ c).$

de Groote treats verbs as taking generalized quantifiers as arguments, as opposed to individuals (in contrast to nouns). This decision is orthogonal to the question of interest here, and requires a richer setup than present in his **2006** paper to treat the composition of semantic translations.

While **de Groote** [2006] does not deal explicitly with sentential connectives, his later work (as described by **Lebedeva** [2012]) does, and is as follows.

Definition *and_dg_tr* : $\omega \rightarrow \omega \rightarrow \omega :=$
 $\text{fun } (P \ Q : \omega) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow P \ c \ (\text{fun } d \Rightarrow Q \ d \ \phi).$

Definition *not_dg_tr* : $\omega \rightarrow \omega :=$
 $\text{fun } (P : \omega) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow (\neg (P \ c \ (\text{fun } _ \Rightarrow \top))) \wedge (\phi \ c).$

Definition *or_dg_tr* : $\omega \rightarrow \omega \rightarrow \omega :=$
 $\text{fun } (P \ Q : \omega) \Rightarrow \text{not_dg_tr } (\text{and_dg_tr } (\text{not_dg_tr } P) (\text{not_dg_tr } Q)).$

Definition *imply_dg_tr* : $\omega \rightarrow \omega \rightarrow \omega :=$
 $\text{fun } (P \ Q : \omega) \Rightarrow \text{not_dg_tr } (\text{and_dg_tr } P (\text{not_dg_tr } Q)).$

Definition *some_dg_tr* : $(e \rightarrow \omega) \rightarrow \omega :=$
 $\text{fun } (P : e \rightarrow \omega) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow \text{some } (\text{fun } (x : e) \Rightarrow P \ x \ (x :: c) \ \phi).$

Definition *every_dg_tr* : $(e \rightarrow \omega) \rightarrow \omega :=$
 $\text{fun } (P : e \rightarrow \omega) \Rightarrow \text{not_dg_tr } (\text{some_dg_tr } (\text{fun } x \Rightarrow \text{not_dg_tr } (P \ x))).$

Definition *pro_dg_tr* : $(e \rightarrow \omega) \rightarrow \omega :=$
 $\text{fun } (P : e \rightarrow \omega) (c : \gamma) (\phi : \gamma \rightarrow t) \Rightarrow P \ (\text{sel } c) \ c \ \phi.$

The function *dg_tr* (short for ‘De Groote’ translation), makes the implicit dynamism and context manipulation in terms explicit.

Fixpoint *dg_tr* ($f : t$) : $\omega :=$
 $\text{match } f \text{ with}$
 $\quad | \top \Rightarrow \text{fun } c \ \phi \Rightarrow \phi \ c$
 $\quad | \perp \Rightarrow \text{fun } _ \Rightarrow \perp$
 $\quad | P \wedge Q \Rightarrow \text{and_dg_tr } (\text{dg_tr } P) (\text{dg_tr } Q)$
 $\quad | P \vee Q \Rightarrow \text{or_dg_tr } (\text{dg_tr } P) (\text{dg_tr } Q)$
 $\quad | P \rightsquigarrow Q \Rightarrow \text{imply_dg_tr } (\text{dg_tr } P) (\text{dg_tr } Q)$
 $\quad | \neg P \Rightarrow \text{not_dg_tr } (\text{dg_tr } P)$
 $\quad | \text{some } P \Rightarrow \text{some_dg_tr } (\text{fun } x \Rightarrow \text{dg_tr } (P \ x))$
 $\quad | \text{all } P \Rightarrow \text{every_dg_tr } (\text{fun } x \Rightarrow \text{dg_tr } (P \ x))$
 $\quad | p \ n \ x \Rightarrow \text{pred_dg_tr } n \ x$

```

| r n x y ⇒ rel_dg_tr n x y
| pro P ⇒ pro_dg_tr (fun x ⇒ dg_tr (P x))
end.

```

Example *if_a_farmer_owns_a_donkey_he_beats_it* : $\forall c \phi$,
 $dg_tr ((some (fun x ⇒ farmer x \wedge some (fun y ⇒ donkey y \wedge own y x))) \rightsquigarrow pro (fun u ⇒ pro (fun v ⇒ beat v u))) c \phi$
 \equiv
 $((all (fun x : e ⇒ farmer x \rightsquigarrow all (fun y : e ⇒ (donkey y \wedge own y x) \rightsquigarrow beat (sel (y :: x :: c)) (sel (y :: x :: c))))))$
 $\wedge \phi c$.

6 The equivalence

The translations must be restricted so as to use the same type γ and same accessor functions *update* and *sel*.

Definition $g_trans := g_tr \ \gamma \ update \ sel$.

Definition $dg_trans := dg_tr \ \gamma \ update \ sel$.

In order to prove the equivalence between de Groote's translation and the composition of the gamma and dynamic translations, I assume that functions are extensional; that two functions are identical iff they compute the same input-output relation.

Hypothesis $fun_ext : \forall (A B : Set) (f g : A \rightarrow B), (\forall x, f x = g x) \rightarrow f = g$.

The main theorem states that, for any formula ϕ of type t , the context-change potential of the contextification of its dynamic translation is identical to the context-change potential of de Groote's translation of it.

Theorem $dg_g_d :$

$\forall (\phi \psi : t), g_trans ((d_tr \ \phi) \ \psi) = fun \ c \Rightarrow (dg_trans \ \phi) \ c (g_trans \ \psi)$.

Proof. This is proven by induction on ϕ , followed by simple but tedious case analysis. \square

One peculiarity about the statement of the theorem concerns the function g_trans on the right hand side of the equation. This is because the discourse context desired on the left hand side (ψ) is of type t , whereas that desired on the right ($g_trans \ \psi$) is of type $\gamma \rightarrow t$. In order for the theorem to make sense, both sides must be given the same discourse continuation. The term g_trans on the right of the equation coerces a term of type t into one of type $\gamma \rightarrow t$. In essence, the theorem says only that g_trans composed with d_tr has the same effect on a discourse configuration of type t that dg_trans has on its image under g_trans .

In practice, this is not much of a restriction, as for any discourse ϕ , the theorem implies that both left and right sides are interpreted identically in the empty discourse continuation.

Corollary *dg_g_d_true* :

$$\forall (\phi : t), g_trans (d_tr \phi \top) = \mathbf{fun} \ c \Rightarrow dg_trans \ \phi \ c (\mathbf{fun} _ \Rightarrow \top).$$

7 Conclusion

The simple perspective offered by [de Groot](#) [2006] on dynamism and dynamic updating has been shown to be decomposable into two simpler and logically independent parts. His 2006 system, modeled here, involves an mapping of *second-order* terms in a source language into those in a target language. This amounts to a tree-to-term homomorphism, and allows [de Groot](#) to avoid defining λ -homomorphisms. This has been improved upon in later work [[Lebedeva, 2012](#)], where the source language is made higher order. In order to have a simple statement of the two parts that De Groot’s system can be factored into (dynamism and context-sensitivity), I have departed from his original presentation in some ways. In particular, I have presented the source and target languages as one and the same, whereas [de Groot](#) treats the source language as properly syntactic, and the target language as properly semantic. Collapsing source and target languages has the aesthetically unpleasant consequence of introducing a ‘meaningless’ term, *pro*, into the language. While the same could have been done with the term **who**, I have chosen to leave it out of the language altogether, as it plays no illustrative semantic role. The second difference concerns the type of verbal elements in the source language, where, for [de Groot](#), they take generalized quantifier denoting expressions as arguments: $\llbracket \mathbf{beat} \rrbracket := \lambda X, Y. X(\lambda x. Y(\lambda y. \mathbf{beat} \ x \ y))$. This is made very simple in his presentation by treating **beat** as a constant in a language different from that of *beat*. This is not available to me, as both *g_tr* and *d_tr* should be defined independently of each other, and therefore must operate on the same source language, whence at least *d_tr* must map one language to itself. While this is a salient theoretical decision regarding the proper treatment of quantifier scope, it does not matter for the decomposition of [de Groot](#)’s translation into the two simpler ones given here.

References

- P. de Groot. Towards a montagovian account of dynamics. In M. Gibson and J. Howell, editors, *Proceedings of SALT 16*, pages 1–16, 2006.
- J. Groenendijk and M. Stokhof. Dynamic predicate logic. *Linguistics and Philosophy*, 14: 39–100, 1991.
- E. Lebedeva. *Expression de la dynamique du discours à l’aide de continuations*. PhD thesis, Université de Lorraine, 2012.